
Read binary tape created by punchdrum and write on drum.
]
i=255
_b a20,e20;
a1=600
     zq                    ; stop; _c_o_m_m_e_n_t ready for binout;
     vy  0                 ; [256] select(0);
a0:  grn 1023  , hs  e0    ; next track: checksum:=0; R:=getchar;
     ga  a2    , hs  e0    ; trackno:=R; R:=getchar;
     vk  0     , pp  40    ; wait for drum;
     ca  127   , zq        ; _i_f R=127 _t_h_e_n stop;
     tk  7     , ac  a2    ; trackno:=trackno+R _s_h_i_f_t 7;
                           ; _f_o_r p:=39 _s_t_e_p −1 _u_n_t_i_l 0 _d_o
a3:  pmn 1.3   DX      IZA ; _b_e_g_i_n R:=1.3; OA:=_t_r_u_e;
a4:  tl  −7    , gr  a5    ; loop: R:=R _s_h_i_f_t −7; a5:=R;
     hs  e0                ;    R:=getchar;
     ga  a6                ;    a6addr:=Raddr;
     ar  a5    X           ;    R:=R+a5; swap RM;
a6:  pi  _0             LZA ;    _i_f OA _t_h_e_n _b_e_g_i_n OA:=_f_a_l_s_e; RC:=marks _e_n_d;
     hv  a4    X       LZ  ;    _i_f R|=0 _t_h_e_n _b_e_g_i_n swap RM; _g_o_t_o loop _e_n_d;
     pp  p−1   X           ;
     tl  3                 ;    
     gr  pa1           MRC ;    buffer[p]:=R+marksRC;
     bs  p     , hv  a3    ; _e_n_d for p;
     lyn a5    , tk −10    ; R:=lyn; R:=R _s_h_i_f_t −10;
     pm  a7                ; _i_f R13−19 |= cell[1023].13−19 _t_h_e_n
     cm  1023  , zq        ; stop; 
a2:  vk  _0     , sk  a1   ; select track; write track;
     hv  a0                ; _g_o_t_o next track;
a5:  qq
a7:  qq  127.19            ; Mask
                           ; _i_n_t_e_g_e_r _p_r_o_c_e_d_u_r_e getchar; _b_e_g_i_n
e0:  lyn e1                ; e1:=lyn;
     gr  e1                ; _c_o_m_m_e_n_t If a character with bad parity
                           ; is read, press Reset, correct char in R, and
                           ; press Normal Start;
     tk  −10   , ac  1023  ; checksum:=checksum+R _s_h_i_f_t −10;
     arn e1    , hr  s1    ; R:=e1; getchar:=R
                           ; _e_n_d getchar;
e1:  qq                    ; character read
_e255;


