

slip<  
[2011-03-22 MK

Program drumtext.

Print out text strings found on the drum.

Text strings are formatted as follows:

The first four bits contain 1111 in all cells but the last. The last cell has 1010 in bits 0-3. The rest of the cell contains up to six characters, each in 6 bits, stored from the right. CARRET is represented by the value 63. Unused character positions in the last cell is filled with the value 10.

The algorithm works as follows:

Read each track on the drum. If the first cell contains 1111 in the leftmost bit positions continue reading until we meet 1010. If we meet any other bit combination in the first four bits, proceed to the next track. If the text pattern is matched, go back to the first track again and punch the text.

```
]
b a20
i=255
al=600                                ; Input buffer
    zq                                ; stop; comment ready for binout;
    vy 32                             ; [256] select(32);
a0: vk -1 t 1                          ; next track:
    bs (a0) t 319                      ; if track no>319 then
    zq                                ; stop
    lk al , vk (a0)                   ; from drum; wait for drum;
    arn al                             ; R:=buffer[0];
    mb 15.3 D                         ; R:=R ^ 4 15 36 0;
    nc 15.3 , hv a0                   ; if R ≠ 4 15 36 0 then goto next track;
    gk a3 , pp 0                      ; save track:=tk; p:=0;
a4: arn pa1                           ; next cell: R:=buffer[p];
    mb 15.3 D                         ; R:=R ^ 4 15 36 0;
    ca 10.3 , hv a5                   ; if R = 4 10 36 0 then goto finish string;
    nc 15.3 , hv a0                   ; if R ≠ 4 15 36 0 then goto next track;
    pp p1 , bs p472                  ; p:=p+1; if p≤39 then
    hv a4                             ; goto next cell;
    bs (a0) t 318                     ; if track no>318 then goto finish string;
    hv a5
    vk (a0) t 1                      ; select next track;
    lk al , vk (a0)                   ; from drum; wait for drum;
    pp 0 , hv a4                     ; p:=0; goto next cell;
a7: m 10-3
    10
a5: arn a3 , tk 10                    ; finish string: R:=save track;
    ga a6                             ;
    tk -30 X
    qq (a6) t -1                      ;
    sy 64 , sy 60                     ; writecr; writechar(UC);
    sy 19 , sy 58                     ; writechar(T); writechar(LC);
    sy 41 , sy 49                     ; writechar(r); writechar(a);
    sy 51 , sy 34                     ; writechar(c); writechar(k);
    sy 60 , sy 59                     ; writechar(UC); writechar(:);
    sy 58 , sy 0                      ; writechar(LC); writechar(SP);
a8: mln a7 , pa a11                   ; write integer from GA4 error track;
a9: ck -10 , bs (a12)                ;
a10: pa a11 t 16                      ;
a11: arn 0 D LZ                       ;
    ga r1 , nc 0                     ;
    sy 0                             ;
a12: ncn 0 t -256                     ;
    mln la7 , hv a9
```

```

sy 64 ; writecr;
a6: vk 0 t 1 ; next track 2:
bs (a6) t 319 ; if track no>319 then
zq ; stop
lk a1 , vk (a6) ; from drum; wait for drum;
pp 0 ; p:=0;
a13: pmn pa1 X ; R:=buffer[p]; M:=0;
cl 34 ; RM:=RM shift 34;
a16: ck -4 , ga a14
ca 10 , hv a0 ; if Raddr=10 then goto next track;
ca 15 , hv a15 ; if Raddr=15 then goto next cell;
ca 63 , it 1 ; if Raddr=63 then out:=64;
a14: sy 0 , cln -6
hv a16
a15: pp p1 , bs p472 ; p:=p+1; if p<39 then
hv a13 ; goto next cell;
hv a6 ; goto next track 2;
a3: qq 0 , qq 0 ; track no of start of text string
e
e255;

```